Adventures in Arrays: A Beginning Tutorial
Jane Stroupe, SAS Institute, Chicago, IL

ABSTRACT
Do you need to
• perform repetitive calculations on a group of variables
• create many variables with the same attributes
• restructure your data
• perform a table lookup with one or more factors?
If your answer is “YES” to any of these questions, you should consider using an array in your DATA step.

INTRODUCTION
In SAS®, an array is simply a way to refer to a group of variables in one observation with a single name.

Think of an array as being an alias for the names of SAS variables on which you need to perform the same task. The variables can be existing variables, new variables or temporary data elements as long as they are the same type.

This tutorial explores the many uses of an array by using examples involving vacation data, collected from ten fictitious vacation resorts. Let's explore the costs and benefits of a vacation at each of these locations.

DATA:
SASDATA.EXPENSES stores off-season rates from ten resorts for room, food, and various activities.

<table>
<thead>
<tr>
<th>ResortName</th>
<th>Resort</th>
<th>Expense1</th>
<th>Expense2</th>
<th>Expense3</th>
<th>Expense4</th>
<th>Expense5</th>
<th>Expense6</th>
</tr>
</thead>
<tbody>
<tr>
<td>Mouse House</td>
<td>HOTEL1</td>
<td>165.89</td>
<td>45.50</td>
<td>78.00</td>
<td>45.00</td>
<td>12.00</td>
<td>35.00</td>
</tr>
<tr>
<td>The Princess Palace</td>
<td>HOTEL2</td>
<td>215.32</td>
<td>64.00</td>
<td>54.00</td>
<td>50.00</td>
<td>18.00</td>
<td>30.00</td>
</tr>
<tr>
<td>Minnie Mansion</td>
<td>HOTEL3</td>
<td>178.90</td>
<td>76.00</td>
<td>32.00</td>
<td>49.00</td>
<td>25.00</td>
<td>29.99</td>
</tr>
<tr>
<td>Sleeping Beauty's</td>
<td>HOTEL4</td>
<td>210.78</td>
<td>54.00</td>
<td>50.00</td>
<td>40.00</td>
<td>17.00</td>
<td>24.00</td>
</tr>
<tr>
<td>Chip and Dale's</td>
<td>HOTEL5</td>
<td>189.87</td>
<td>49.00</td>
<td>67.00</td>
<td>63.00</td>
<td>38.00</td>
<td>42.00</td>
</tr>
<tr>
<td>Cinderella's Castle</td>
<td>HOTEL6</td>
<td>312.15</td>
<td>78.00</td>
<td>86.00</td>
<td>67.00</td>
<td>14.00</td>
<td>19.40</td>
</tr>
<tr>
<td>Dalmation Doghouse</td>
<td>HOTEL7</td>
<td>197.12</td>
<td>45.00</td>
<td>95.00</td>
<td>32.00</td>
<td>26.00</td>
<td>18.95</td>
</tr>
<tr>
<td>Peter Pan's Playhouse</td>
<td>HOTEL8</td>
<td>240.98</td>
<td>43.00</td>
<td>65.13</td>
<td>45.00</td>
<td>12.00</td>
<td>32.00</td>
</tr>
<tr>
<td>Bambi Bed &amp; Breakfast</td>
<td>HOTEL9</td>
<td>312.10</td>
<td>29.98</td>
<td>32.87</td>
<td>24.00</td>
<td>15.00</td>
<td>26.00</td>
</tr>
<tr>
<td>Fantasia Funhouse</td>
<td>HOTEL10</td>
<td>187.98</td>
<td>36.00</td>
<td>21.54</td>
<td>43.00</td>
<td>24.00</td>
<td>29.65</td>
</tr>
</tbody>
</table>

EXAMPLE 1: USING A ONE-DIMENSIONAL ARRAY TO PERFORM REPETITIVE CALCULATIONS.

TASK:
You want a SAS data set with both the off-season and seasonal rates. The seasonal rates are 25% higher than the off-season rates.

SOLUTIONS:
• Use six assignment statements
• Use an array.
PROGRAM – VERSION 1:
The first solution requires considerable typing:
```
data work.SeasonRates;
  set sasdata.expenses(drop=ResortName);
  seasonal1 = expense1*1.25;
  seasonal2 = expense2*1.25;
  seasonal3 = expense3*1.25;
  seasonal4 = expense4*1.25;
  seasonal5 = expense5*1.25;
  seasonal6 = expense6*1.25;
run;
```

PROGRAM – VERSION 2:
To solve this problem using an array, you must first establish the correspondence between
- one array name and the existing variables, expense1 - expense6
- a second array name and the new variables, seasonal1 - seasonal6.

Use array statements to set up the correspondences:
```
array ex{6} expense1 - expense6;
array seasonal{6} seasonal1 - seasonal6;
```

The ARRAY statement is a compile time statement that creates the correspondence between the array name and, in this example, the variables in the Program Data Vector.

The general form of a simple ARRAY statement
```
ARRAY array-name{n} $ length elements;
array-name any valid SAS name (avoid function names)
{n} number of elements to which the array refers (often referred to as the dimension(s) of the array)
$ indication that the array refers to new character data variables
length the length for new character variables. The default is 8.
elements the variables in the data to which the array refers. If the variables do not exist, SAS will create them.
```

Once you establish this correspondence, you can use a DO loop to execute six assignment statements that calculate the weekly total.
```
do i=1 to 6;
  seasonal{i} = ex{i}*1.25;
end;
```
Putting this all together in a second version of the DATA step:

```sas
data work.SeasonRates(drop=i);
  set sasdata.expenses(drop=ResortName);
  array ex{6} expense1 - expense6;
  array seasonal{6} seasonal1 - seasonal6;
  do i=1 to 6;
    seasonal{i} = ex{i}*1.25;
  end;
  format expense1 - expense6 seasonal1 - seasonal6 dollar9.2;
run;
```

1. To use an array, you must first declare it in an ARRAY statement. At compile time, the ARRAY statement establishes a correspondence between the name of the array, EX, and the numeric DATA step variables to which it refers, EXPENSE1 - EXPENSE6.

The association of an array reference with a variable is only for the duration of the DATA step. If you need to refer to the array in another DATA step, you must repeat the array reference in that DATA step.

2. The array, SEASONAL, references variables that do not exist in the Program Data Vector. SAS creates the variables SEASONAL1 - SEASONAL6 from the ARRAY statement.

3. The DO loop processes each element of the arrays sequentially. The assignment statement using the array reference calculates seasonal rates for each of the six expenses.

4. Any FORMAT, LABEL, DROP, KEEP, or LENGTH statements must use DATA step variable names, not array references.

Data Set SeasonRates (8 of 14 variables)

<table>
<thead>
<tr>
<th>ResortName</th>
<th>Resort</th>
<th>seasonal1</th>
<th>seasonal2</th>
<th>seasonal3</th>
<th>seasonal4</th>
<th>seasonal5</th>
<th>seasonal6</th>
</tr>
</thead>
<tbody>
<tr>
<td>Mouse House</td>
<td>HOTEL1</td>
<td>$207.36</td>
<td>$56.88</td>
<td>$97.50</td>
<td>$56.25</td>
<td>$15.00</td>
<td>$43.75</td>
</tr>
<tr>
<td>The Princess Palace</td>
<td>HOTEL2</td>
<td>$269.15</td>
<td>$80.00</td>
<td>$67.50</td>
<td>$62.50</td>
<td>$22.50</td>
<td>$37.50</td>
</tr>
<tr>
<td>Minnie Mansion</td>
<td>HOTEL3</td>
<td>$223.63</td>
<td>$95.00</td>
<td>$40.00</td>
<td>$61.25</td>
<td>$31.25</td>
<td>$37.49</td>
</tr>
<tr>
<td>Sleeping Beauty's Reststop</td>
<td>HOTEL4</td>
<td>$263.48</td>
<td>$67.50</td>
<td>$62.50</td>
<td>$50.00</td>
<td>$21.25</td>
<td>$30.00</td>
</tr>
<tr>
<td>Chip and Dale's Nuthouse</td>
<td>HOTEL5</td>
<td>$237.34</td>
<td>$61.25</td>
<td>$83.75</td>
<td>$78.75</td>
<td>$47.50</td>
<td>$52.50</td>
</tr>
<tr>
<td>Cinderella's Castle Inn</td>
<td>HOTEL6</td>
<td>$390.19</td>
<td>$97.50</td>
<td>$107.50</td>
<td>$83.75</td>
<td>$17.50</td>
<td>$24.25</td>
</tr>
<tr>
<td>Dalmation Doghouse</td>
<td>HOTEL7</td>
<td>$246.40</td>
<td>$56.25</td>
<td>$118.75</td>
<td>$40.00</td>
<td>$32.50</td>
<td>$23.69</td>
</tr>
<tr>
<td>Peter Pan's Playhouse</td>
<td>HOTEL8</td>
<td>$301.23</td>
<td>$53.75</td>
<td>$81.41</td>
<td>$56.25</td>
<td>$15.00</td>
<td>$40.00</td>
</tr>
<tr>
<td>Bambi Bed &amp; Breakfast</td>
<td>HOTEL9</td>
<td>$390.13</td>
<td>$37.48</td>
<td>$41.09</td>
<td>$30.00</td>
<td>$18.75</td>
<td>$32.50</td>
</tr>
<tr>
<td>Fantasia Funhouse</td>
<td>HOTEL10</td>
<td>$234.98</td>
<td>$45.00</td>
<td>$26.93</td>
<td>$53.75</td>
<td>$30.00</td>
<td>$37.06</td>
</tr>
</tbody>
</table>
Valid Syntax for the ARRAY statement:
1. The array name can be the same as a variable list root word as long as there is no variable named the root word. You should, however, avoid function names since array names that are the same as a SAS function turn off the availability of that function for the duration of the DATA step.

   For example, these ARRAY statements are valid because there is not a variable named EXPENSE in the data set.
   array expense{6} expense1 - expense6;
   or
   array expense{6};

2. The variable names do not have to be a variable list.

   If the data set variables were named ROOMRATE, DAILYFOOD, SPAVISITS, ROUNDOFGOLF, HORSEBACKRIDING, and PARKADMISSION, you could use the following ARRAY statement.
   array expense{6} RoomRate DailyFood SpaVisits RoundOfGolf HorseBackRiding ParkAdmission;

   You can use the double dash to refer to this list of variables that are contiguous in the SAS data set.
   array expense{6} RoomRate -- ParkAdmission;

   You could refer to the variable list, just not in ascending suffix order.
   array expense{6} expense6 expense3 expense2 expense4 expense5 expense1;

   Since EXPENSE1 - EXPENSE6 are all numeric variables, you can use the key word _NUMERIC_ to refer to them. You can use the _NUMERIC_ keyword only for variables currently defined in the PDV.
   array expense{6} _numeric_;

   For the new variables, you can name the variables with a list that has a different root word.

   For example:
   array seasonalexpense{6} seasonal1 - seasonal6;
   or
   array seasonal{6} sRoomRate sDailyFood sSpaVisits sRoundOfGolf sHorseBackRiding sParkAdmission;
   or
   array seasonalexpense{6} seasonal6 seasonal4 seasonal2 seasonal1 seasonal5 seasonal3;

   You can not use the double dash to refer to new variable names.

3. The ARRAY statement can automatically name new variables.

   For example, this ARRAY statement automatically creates variables SEASONAL1 - SEASONAL6 since they do not exist in the Program Data Vector.
   array seasonal{6};
4. You do not have to specify the dimensions with a number as we did in the array statements. However, during compile time, SAS has to know what the dimensions are. You can use a macro variable to specify the dimensions. For example:

```sas
%let number=6;
array ex{&number} expense1 – expense6;
```

If appropriate, you can use a range of values for the dimensions. For example, use this ARRAY statement and DO loop to process the range of values.

```sas
array ex{94:99} expense94 – expense99;
do i=94 to 99;
  SAS statements
end;
```

Sometimes, you can use the * to specify the dimensions and the DIM function in your DO loop.

```sas
array ex{*} expense1 – expense6;
do i=1 to dim(ex);
  SAS statements
end;
```

If you use the *, you must specify the elements or variables. The DIM function takes an array name as its argument and returns the dimension of the array.

**EXAMPLE 2: USING A ONE-DIMENSIONAL ARRAY TO PERFORM A TABLE LOOKUP.**

**TASK:**
You have budgeted the following amounts for each of the daily expenses: $175 for expense1 (room), $75 for expense2 (food), $25 for expense3 (spa treatments), $35 for expense4 (a round of golf), $25 for expense5 (horseback riding), and $30 for expense6 (theme park admission). Use an array to assign these budget amounts as initial values and determine the difference between the budgeted amounts and the actual rates.

**SOLUTIONS:**
- Use six assignment statements to calculate the differences.
- Use an array.

**PROGRAM**

```sas
data work.diffs;
drop i;
set sasdata.expenses
  (drop=ResortName);
array budget{6} _temporary_ (175,75,25,35,25,30);
array expense{*} expense1 – expense6;
array diff{6};
do i=1 to dim(expense);
  diff{i} = budget{i} - expense{i};
end;
run;
```

You can assign initial values for the corresponding elements in the array by putting these initial values within parenthesis separated by either a comma or a blank. Initial values can be assigned to both variables and temporary data elements.

In the ARRAY statement, the keyword _TEMPORARY_ creates a list of temporary data elements that can be either numeric or character. The temporary elements must be used in a DATA step and are automatically retained for each iteration of the DATA step. You must refer to these elements by the array name and dimension since they do not have variable names, and you cannot use the asterisk (*) to refer to all the elements.

Arrays of temporary elements are useful when the only purpose for creating an array is to perform a calculation (often referred to as performing a table lookup). You can improve performance CPU time by using temporary array references.
You can use the DIM function to return the dimensions of the array EXPENSE.

### Data Set Diffs (7 of 13 variables)

<table>
<thead>
<tr>
<th>Resort</th>
<th>diff1</th>
<th>diff2</th>
<th>diff3</th>
<th>diff4</th>
<th>diff5</th>
<th>diff6</th>
</tr>
</thead>
<tbody>
<tr>
<td>HOTEL1</td>
<td>9.11</td>
<td>29.50</td>
<td>-53.00</td>
<td>-10</td>
<td>13</td>
<td>-5.00</td>
</tr>
<tr>
<td>HOTEL2</td>
<td>-40.32</td>
<td>11.00</td>
<td>-29.00</td>
<td>-15</td>
<td>7</td>
<td>0.00</td>
</tr>
<tr>
<td>HOTEL3</td>
<td>-3.90</td>
<td>-1.00</td>
<td>-7.00</td>
<td>-14</td>
<td>0</td>
<td>0.01</td>
</tr>
<tr>
<td>HOTEL4</td>
<td>-35.78</td>
<td>21.00</td>
<td>-25.00</td>
<td>-5</td>
<td>8</td>
<td>6.00</td>
</tr>
<tr>
<td>HOTEL5</td>
<td>-14.87</td>
<td>26.00</td>
<td>-42.00</td>
<td>-28</td>
<td>-13</td>
<td>-12.00</td>
</tr>
<tr>
<td>HOTEL6</td>
<td>-137.15</td>
<td>-3.00</td>
<td>-61.00</td>
<td>-32</td>
<td>11</td>
<td>10.60</td>
</tr>
<tr>
<td>HOTEL7</td>
<td>-22.12</td>
<td>30.00</td>
<td>-70.00</td>
<td>3</td>
<td>-1</td>
<td>11.05</td>
</tr>
<tr>
<td>HOTEL8</td>
<td>-65.98</td>
<td>32.00</td>
<td>-40.13</td>
<td>-10</td>
<td>13</td>
<td>-2.00</td>
</tr>
<tr>
<td>HOTEL9</td>
<td>-137.10</td>
<td>45.02</td>
<td>-7.87</td>
<td>11</td>
<td>10</td>
<td>4.00</td>
</tr>
<tr>
<td>HOTEL10</td>
<td>-12.98</td>
<td>39.00</td>
<td>3.46</td>
<td>-8</td>
<td>1</td>
<td>0.35</td>
</tr>
</tbody>
</table>
EXAMPLE 3: USING A ONE-DIMENSIONAL ARRAY TO RESTRUCTURE YOUR DATA AND PERFORM A TABLE LOOKUP WITH CHARACTER DATA.

TASK:
You need to create the following graph.

In order to create the graph, the data must be in this form.

<table>
<thead>
<tr>
<th>Data Set Rotate</th>
</tr>
</thead>
<tbody>
<tr>
<td>Resort</td>
</tr>
<tr>
<td>HOTEL1</td>
</tr>
<tr>
<td>HOTEL1</td>
</tr>
<tr>
<td>HOTEL1</td>
</tr>
<tr>
<td>HOTEL1</td>
</tr>
<tr>
<td>HOTEL1</td>
</tr>
<tr>
<td>HOTEL1</td>
</tr>
<tr>
<td>HOTEL2</td>
</tr>
<tr>
<td>HOTEL2</td>
</tr>
<tr>
<td>HOTEL2</td>
</tr>
</tbody>
</table>

SOLUTIONS:
- Use complicated logic to rotate the data, then use 6 assignment statements to create a variable that is the name of the expense.
- Use PROC TRANSPOSE to rotate the data, then a DATA step with assignment statements to create a variable that is the name of the expense.
- Use an array.
**PROGRAM**

```sas
data work.rotate;
  keep resort expense amount TypeOfExpense;
  set sasdata.expenses;
  array ex{*} _numeric_
  array name{6} $ 32 ('Room Rate'
    'Daily Food Allowance'
    'Round of Golf'
    'Spa Visit'
    'Horse Back Riding'
    'Theme Park Admission');
  do Expense = lbound(ex) to hbound(ex);
    Amount = ex{Expense};
    TypeOfExpense = name{Expense};
    output;
  end;
run;
```

1. You can use the following SAS variables to reference variables that have been previously defined in the same DATA step:
   - `_NUMERIC_` indicates all numeric variables.
   - `_CHARACTER_` indicates all character variables.
   - `_ALL_` indicates all variables. For arrays, all the previously defined variables must be of the same type.

   In this case `_NUMERIC_` refers to the variables `EXPENSE1` – `EXPENSE6`. The SET statement must come before the ARRAY statement in order to refer to these variables with the `_NUMERIC_` keyword.

2. Since the array `NAME` refers to character values, the ARRAY statement uses the `$` to indicate character data and a length (32). The length needs to be established in the array statement; otherwise, SAS uses the default length of 8. You can use a length statement prior to the array statement to assign the length.

3. The index variable, `EXPENSE`, becomes a data set variable that, for this example, contains the number of the expense, 1 - 6.

   You can use the HBOUND function to return the upper bound of a one-dimensional array and the LBOUND function to return the lower bound. By using these functions you can avoid changing the bounds of an iterative DO loop each time you change the bounds of the array. The HBOUND and LBOUND functions are especially useful if you are using a range for the dimensions.

   For example:
   ```sas
   array expense{100:105} expense1 - expense6;
   do i = lbound(expense) to hbound(expense);
   ```

4. Assign the array value in EX{expense} to the variable `AMOUNT`.

5. Assign the array value in NAME{expense} to the variable `TYPEOFEXPENSE`.

6. Use an OUTPUT statement in the DO loop in order to write out an observation for each iteration of the loop.
EXAMPLE 4: USING A TWO-DIMENSIONAL ARRAY

You can think of a two-dimensional array as having 'rows' and 'columns', but to SAS, a two-dimensional array is no different than a one-dimensional array. It is still an alias for data set variables in the PDV or for temporary data elements.

TASK:
You know the following information about daily rates for taxes and gratuity charged by the resorts.

<table>
<thead>
<tr>
<th>Resort</th>
<th>Taxes</th>
<th>Gratuity</th>
</tr>
</thead>
<tbody>
<tr>
<td>1</td>
<td>18.54</td>
<td>14</td>
</tr>
<tr>
<td>2</td>
<td>17.84</td>
<td>20</td>
</tr>
<tr>
<td>3</td>
<td>12.50</td>
<td>15</td>
</tr>
<tr>
<td>4</td>
<td>14.25</td>
<td>18</td>
</tr>
<tr>
<td>5</td>
<td>16.33</td>
<td>12</td>
</tr>
<tr>
<td>6</td>
<td>19.00</td>
<td>16</td>
</tr>
</tbody>
</table>

...:

You want to combine this information with the data in SASDATA.EXPENSES to calculate the total cost for each resort.

SOLUTIONS:
- Create a SAS data set containing these constants and merge it with SASDATA.EXPENSES.
- Use IF..THEN logic and assignment statements to create the new variables.
- Since this information is in rows and columns, use a temporary two-dimensional array to hold the tax and gratuity information.

PROGRAM
```
data work.total;
  drop num;
  array charge{10,2} _temporary_
    (18.54      14
     17.84      20
     12.50      15
     14.25      18
     16.33      12
     19.00      16
     12.75      19
     14.98      16
     15.76      20
     13.75      17);
  set sasdata.expenses;
  num = input(substr(resort,6),2.);
  tax = charge{num,1};
  gratuity = charge{num,2};
  total = sum(of expense1-expense6,tax,gratuity);
run;
```

The array, CHARGE, refers to 10 rows and 2 columns. It is used to hold constants. Even though you can type them into rows and columns and that helps you visualize what the program needs to do, SAS is assigning these values to contiguous slots in memory. You can refer to these slots by row and column number.

<table>
<thead>
<tr>
<th>charge{1,1}</th>
<th>charge{1,2}</th>
<th>charge{2,1}</th>
<th>charge{2,2}</th>
<th>charge{3,1}</th>
<th>...</th>
</tr>
</thead>
<tbody>
<tr>
<td>18.54</td>
<td>14</td>
<td>17.84</td>
<td>20</td>
<td>12.50</td>
<td></td>
</tr>
</tbody>
</table>
If you had created variables instead of using _TEMPORARY_, the variables would have been named
CHARGE1 referring to CHARGE{1,1}
CHARGE2 referring to CHARGE{1,2}
CHARGE3 referring to CHARGE{2,1}
CHARGE4 referring to CHARGE{2,2},
and so forth.

The variable RESORT in SASDATA.EXPENSES has values HOTEL1, HOTEL2, etc. You want to use this as an
index when you reference the array, but an array index must be numeric. Therefore, you must extract the digit
from the end of this value by using the SUBSTR('text',start,end) function. The SUBSTR function returns a
character value. Using the INPUT function then converts the character value to numeric.

Create a variable TAX and a variable GRATUITY using the appropriate resort number as the row dimension, and
the appropriate number as the column dimension, where column 1 is the value for TAX and column 2 is the value
for GRATUITY.

<table>
<thead>
<tr>
<th>ResortName</th>
<th>Resort</th>
<th>Expense1</th>
<th>Expense2</th>
<th>Expense3</th>
<th>Expense4</th>
<th>Expense5</th>
<th>Expense6</th>
<th>tax</th>
<th>gratuity</th>
<th>total</th>
</tr>
</thead>
<tbody>
<tr>
<td>Mouse House</td>
<td>HOTEL1</td>
<td>165.89</td>
<td>45.50</td>
<td>78.00</td>
<td>45.00</td>
<td>12.00</td>
<td>35.00</td>
<td>18.54</td>
<td>14</td>
<td>413.93</td>
</tr>
<tr>
<td>The Princess Palace</td>
<td>HOTEL2</td>
<td>215.32</td>
<td>64.00</td>
<td>54.00</td>
<td>50.00</td>
<td>18.00</td>
<td>30.00</td>
<td>17.84</td>
<td>20</td>
<td>469.16</td>
</tr>
<tr>
<td>Minnie Mansion</td>
<td>HOTEL3</td>
<td>178.90</td>
<td>76.00</td>
<td>32.00</td>
<td>49.00</td>
<td>25.00</td>
<td>29.99</td>
<td>12.50</td>
<td>15</td>
<td>418.39</td>
</tr>
<tr>
<td>Sleeping Beauty's</td>
<td>HOTEL4</td>
<td>210.78</td>
<td>54.00</td>
<td>50.00</td>
<td>40.00</td>
<td>17.00</td>
<td>24.00</td>
<td>14.25</td>
<td>18</td>
<td>428.03</td>
</tr>
<tr>
<td>Chip and Dale's</td>
<td>HOTEL5</td>
<td>189.87</td>
<td>49.00</td>
<td>67.00</td>
<td>63.00</td>
<td>38.00</td>
<td>42.00</td>
<td>16.33</td>
<td>12</td>
<td>477.20</td>
</tr>
<tr>
<td>Nuthouse</td>
<td>HOTEL6</td>
<td>312.15</td>
<td>78.00</td>
<td>86.00</td>
<td>67.00</td>
<td>14.00</td>
<td>19.40</td>
<td>19.00</td>
<td>16</td>
<td>611.55</td>
</tr>
<tr>
<td>Cinderella's Castle</td>
<td>HOTEL7</td>
<td>197.12</td>
<td>45.00</td>
<td>95.00</td>
<td>32.00</td>
<td>26.00</td>
<td>18.95</td>
<td>12.75</td>
<td>19</td>
<td>445.82</td>
</tr>
<tr>
<td>Doghouse</td>
<td>HOTEL8</td>
<td>240.98</td>
<td>43.00</td>
<td>65.13</td>
<td>45.00</td>
<td>12.00</td>
<td>32.00</td>
<td>14.98</td>
<td>16</td>
<td>469.09</td>
</tr>
<tr>
<td>Peter Pan's Playhouse</td>
<td>HOTEL9</td>
<td>312.10</td>
<td>29.98</td>
<td>32.87</td>
<td>24.00</td>
<td>15.00</td>
<td>26.00</td>
<td>15.76</td>
<td>20</td>
<td>475.71</td>
</tr>
<tr>
<td>Bambi Bed &amp; Breakfast</td>
<td>HOTEL10</td>
<td>187.98</td>
<td>36.00</td>
<td>21.54</td>
<td>43.00</td>
<td>24.00</td>
<td>29.65</td>
<td>13.75</td>
<td>17</td>
<td>372.92</td>
</tr>
</tbody>
</table>
EXAMPLE 5: CREATING A MULTI-DIMENSIONAL ARRAY FROM A SAS DATA SET.

TASK:
You need to determine the overall ratings for your choice of resorts. The SAS data set, SASDATA.HTLRATE, contains ratings for each hotel in each of the categories. room, food, and the activities offered. Room is rated on a scale from 1 to 5, food is rated on a scale from 1 to 4, and activities is rated on a scale from 1 to 3. The higher the rating, the better the hotel.

<table>
<thead>
<tr>
<th>Hotel</th>
<th>RoomRate</th>
<th>FoodRate</th>
<th>ActRate</th>
</tr>
</thead>
<tbody>
<tr>
<td>HOTEL1</td>
<td>4</td>
<td>3</td>
<td>1</td>
</tr>
<tr>
<td>HOTEL2</td>
<td>2</td>
<td>4</td>
<td>2</td>
</tr>
<tr>
<td>HOTEL3</td>
<td>5</td>
<td>2</td>
<td>1</td>
</tr>
<tr>
<td>HOTEL4</td>
<td>2</td>
<td>4</td>
<td>2</td>
</tr>
<tr>
<td>HOTEL5</td>
<td>4</td>
<td>1</td>
<td>3</td>
</tr>
<tr>
<td>HOTEL6</td>
<td>2</td>
<td>2</td>
<td>3</td>
</tr>
<tr>
<td>HOTEL7</td>
<td>5</td>
<td>3</td>
<td>2</td>
</tr>
<tr>
<td>HOTEL8</td>
<td>2</td>
<td>4</td>
<td>3</td>
</tr>
<tr>
<td>HOTEL9</td>
<td>3</td>
<td>2</td>
<td>3</td>
</tr>
<tr>
<td>HOTEL10</td>
<td>5</td>
<td>3</td>
<td>1</td>
</tr>
</tbody>
</table>

The SAS data set, SASDATA.RATINGS, has descriptions of the overall ratings from a guidebook. You need to combine these in order to determine the best place to stay.

<table>
<thead>
<tr>
<th>ROOM</th>
<th>FOOD</th>
<th>ACTIVITY</th>
<th>RATING</th>
</tr>
</thead>
<tbody>
<tr>
<td>1</td>
<td>1</td>
<td>1</td>
<td>Awful</td>
</tr>
<tr>
<td>1</td>
<td>1</td>
<td>2</td>
<td>Awful</td>
</tr>
<tr>
<td>1</td>
<td>1</td>
<td>3</td>
<td>Maybe</td>
</tr>
<tr>
<td>5</td>
<td>3</td>
<td>2</td>
<td>Good</td>
</tr>
<tr>
<td>5</td>
<td>3</td>
<td>3</td>
<td>Great</td>
</tr>
<tr>
<td>5</td>
<td>4</td>
<td>1</td>
<td>Good</td>
</tr>
<tr>
<td>5</td>
<td>4</td>
<td>2</td>
<td>Great</td>
</tr>
<tr>
<td>5</td>
<td>4</td>
<td>3</td>
<td>Fantastic</td>
</tr>
</tbody>
</table>

PROGRAM:
```
data compare(keep=hotel roomrate foodrate actrate HotelRating);
  array rate{5,4,3} $ _temporary_;  
  if _n_ = 1 then do i=1 to all;
    set sasdata.ratings nobs=all;
    rate{room,food,activity}=rating;
  end;
  set sasdata.htlrate;
  HotelRating=rate{roomrate,foodrate,actrate};
run;
```

The dimensions (5,4,3) indicate that the array, RATE, refers to 5 “pages”, 4 “rows”, and 3 “columns”. In this example, RATE refers to 5 room rates, 4 food rates, and 3 activity rates. If the ARRAY statement had not used the keyword, _TEMPORARY_, SAS would have created 60 variables, RATE1 - RATE60. With the _TEMPORARY_ keyword, RATE refers to 60 contiguous “slots” in memory in which the values of SASDATA.RATINGS will be stored.
The first time through the DATA step (when _n_ = 1), the value of RATING from SASDATA.RATINGS is loaded into the array. During compile time, the NOBS=ALL SET statement option assigns the number of observations from SASDATA.RATINGS (60) to the temporary numeric variable ALL. Therefore, the DO loop will execute 60 times.

ROOM, FOOD, and ACTIVITY are used as the dimensions to load RATING into the correct slot in memory.

After the array is loaded, SASDATA.HTLRATE is read.

The variable, HOTELRATING, is created by assigning the appropriate value of the RATE array.

Data Set Compare

<table>
<thead>
<tr>
<th>Obs</th>
<th>Hotel</th>
<th>RoomRate</th>
<th>FoodRate</th>
<th>ActRate</th>
<th>HotelRating</th>
</tr>
</thead>
<tbody>
<tr>
<td>1</td>
<td>HOTEL1</td>
<td>4</td>
<td>3</td>
<td>1</td>
<td>OK</td>
</tr>
<tr>
<td>2</td>
<td>HOTEL2</td>
<td>2</td>
<td>4</td>
<td>2</td>
<td>OK</td>
</tr>
<tr>
<td>3</td>
<td>HOTEL3</td>
<td>5</td>
<td>2</td>
<td>1</td>
<td>OK</td>
</tr>
<tr>
<td>4</td>
<td>HOTEL4</td>
<td>2</td>
<td>4</td>
<td>2</td>
<td>OK</td>
</tr>
<tr>
<td>5</td>
<td>HOTEL5</td>
<td>4</td>
<td>1</td>
<td>3</td>
<td>OK</td>
</tr>
<tr>
<td>6</td>
<td>HOTEL6</td>
<td>2</td>
<td>2</td>
<td>3</td>
<td>Maybe</td>
</tr>
<tr>
<td>7</td>
<td>HOTEL7</td>
<td>5</td>
<td>3</td>
<td>2</td>
<td>Good</td>
</tr>
<tr>
<td>8</td>
<td>HOTEL8</td>
<td>2</td>
<td>4</td>
<td>3</td>
<td>OK</td>
</tr>
<tr>
<td>9</td>
<td>HOTEL9</td>
<td>3</td>
<td>2</td>
<td>3</td>
<td>OK</td>
</tr>
<tr>
<td>10</td>
<td>HOTEL10</td>
<td>5</td>
<td>3</td>
<td>1</td>
<td>OK</td>
</tr>
</tbody>
</table>

CONCLUSION
If your data processing needs repetitive code, creation of many variables with the same attributes, converting data from a horizontal structure to a vertical structure, assigning many constants, then think ARRAY. Arrays are powerful tools because they allow you to group similar variables under one umbrella alias for similar processing. Though array references are assigned at compile time, the syntax allows for flexibility when declaring variable names and dimensions.

So enjoy yourself as your adventure in arrays takes you further in your data step programming.

ACKNOWLEDGMENTS
The author would like to thank Linda Jolley, Cynthia Zender, Rick Bell, and Kay Alden who helped the author by reviewing this paper.

CONTACT INFORMATION
Your comments and questions are valued and encouraged. Contact the author at:
Jane Stroupe
Phone: 847.367.6216
Email: Jane.Stroupe@sas.com

SAS and all other SAS Institute Inc. product or service names are registered trademarks or trademarks of SAS Institute Inc. in the USA and other countries. © indicates USA registration. Other brand and product names are trademarks of their respective companies.